

ΥΠΟΛΟΓΙΣΤΕΣ Ι

ΜΟΝΟΔΙΑΣΤΑΤΟΙ ΠΙΝΑΚΕΣ

1

Γιατί χρειαζόμαστε πίνακες;

- Σε πολλά προγράμματα μπορεί να χρειαστεί να ορίσουμε πολλές μεταβλητές παρόμοιου τύπου
 - π.χ. να ορίσουμε και σώσουμε τις τιμές μιας συνάρτησης σε N σημεία
- Αντί να δηλώσουμε N διαφορετικές μεταβλητές, δηλώνουμε μία μεταβλητή με N δείκτες → **πίνακας**
- **Πίνακας: μεταβλητή με δείκτες**

Απλή μεταβλητή: ένα όνομα, μία θέση στην μνήμη
Πίνακας: ένα όνομα, πολλές διαδοχικές θέσεις στην μνήμη

Παράδειγμα #1A (χωρίς πίνακες)

Κατασκευάστε πρόγραμμα που θα δέχεται 5 αριθμούς και θα τυπώνει το τετράγωνό τους

```
PROGRAM SQUARE_1A
  IMPLICIT NONE
  DOUBLE PRECISION A1, A2, A3, A4, A5

  WRITE(*,*) 'ΔΩΣΤΕ 5 ΑΡΙΘΜΟΥΣ'
  READ(*,*) A1, A2, A3, A4, A5

  WRITE(*,*) 'ΤΟ ΤΕΤΡΑΓΩΝΟ ΤΟΥ', A1, 'ΕΙΝΑΙ', A1**2
  WRITE(*,*) 'ΤΟ ΤΕΤΡΑΓΩΝΟ ΤΟΥ', A2, 'ΕΙΝΑΙ', A2**2
  WRITE(*,*) 'ΤΟ ΤΕΤΡΑΓΩΝΟ ΤΟΥ', A3, 'ΕΙΝΑΙ', A3**2
  WRITE(*,*) 'ΤΟ ΤΕΤΡΑΓΩΝΟ ΤΟΥ', A4, 'ΕΙΝΑΙ', A4**2
  WRITE(*,*) 'ΤΟ ΤΕΤΡΑΓΩΝΟ ΤΟΥ', A5, 'ΕΙΝΑΙ', A5**2

  END
```

3

Παράδειγμα #1B (με πίνακες)

Κατασκευάστε πρόγραμμα που θα δέχεται 5 αριθμούς και θα τυπώνει το τετράγωνό τους

```
PROGRAM SQUARE_1B
  IMPLICIT NONE
  DOUBLE PRECISION A(5)
  INTEGER I

  WRITE(*,*) 'ΔΩΣΤΕ 5 ΑΡΙΘΜΟΥΣ'
  READ(*,*) A(1), A(2), A(3), A(4), A(5)

  DO I = 1, 5
    WRITE(*,*) 'ΤΟ ΤΕΤΡΑΓΩΝΟ ΤΟΥ', A(I), 'ΕΙΝΑΙ', A(I)**2
  END DO

  END
```

4

Δήλωση μονοδιάστατων πινάκων

- Όλα τα στοιχεία ενός πίνακα είναι υποχρεωτικά του ίδιου τύπου
- Δηλώνεται ως συνήθης μεταβλητή αλλά με μέγεθος
- Το μέγεθος αυτομάτως επισημαίνει ότι είναι πίνακας

Παράδειγμα: δηλώστε σε πρόγραμμα έναν ακέραιο πίνακα 100 στοιχείων και έναν ρητό 200 στοιχείων

```
PROGRAM MATRIX
  IMPLICIT NONE
  INTEGER K(100)
  DOUBLE PRECISION A(200)

  εντολές...

END
```

5

Μέγεθος πίνακα

- Στη δήλωση του πίνακα το μέγεθος πρέπει να είναι μια σταθερή αριθμητική τιμή
 - ακέραιο νούμερο
 - σταθερή ακέραια μεταβλητή με αναθετημένη τιμή
ο μεταφραστής πρέπει να ξέρει πόσο χώρο στην μνήμη χρειάζεται
- Για δημιουργία σταθερής ακέραιας μεταβλητής χρησιμοποιούμε την εντολή **parameter**

τύπος μεταβλητή
parameter(μεταβλητή=τιμή)

- Η ανάθεση γίνεται πριν την δήλωση του πίνακα
- Η τιμή της μεταβλητής δεν μπορεί να αλλάξει στο πρόγραμμα

6

Παράδειγμα #2

Δηλώστε σε πρόγραμμα:

- 3 ακέραιους πίνακες 100 στοιχείων, και
- 3 πίνακες διπλής ακρίβειας 100 στοιχείων

```
PROGRAM MATRIX2
  IMPLICIT NONE
  INTEGER NMAX
  PARAMETER (NMAX=100)
  INTEGER I(NMAX), J(NMAX), K(NMAX)
  DOUBLE PRECISION A(NMAX), B(NMAX), C(NMAX)

  εντολές...

END
```

7

Ανάθεση τιμών σε πίνακα

- Καλούμε τα στοιχεία του πίνακα χρησιμοποιώντας ακέραιους δείκτες
 - προσοχή να μην ξεπεράσουμε τα όρια του πίνακα
- Από εκεί και πέρα όπως κάθε άλλη μεταβλητή

Γενικά μπορούμε να χρησιμοποιήσουμε 3 τρόπους για την γρήγορη ανάθεση τιμών:

- με απευθείας ανάθεση
- με εντολή **DATA**
- με εντολή εισόδου **READ**

8

Παράδειγμα ανάθεσης τιμών #1

Δηλώστε πίνακα ακεραίων με 5 στοιχεία, και αναθέστε τους τις τιμές 10, 20, ..., 50 αντίστοιχα

1. Απευθείας ανάθεση:

```
PROGRAM MATRIX3A
  IMPLICIT NONE
  INTEGER A(5)

  A(1) = 10
  A(2) = 20
  A(3) = 30
  A(4) = 40
  A(5) = 50

END
```

```
PROGRAM MATRIX3A
  IMPLICIT NONE
  INTEGER A(5), I

  DO I = 1, 5
    A(I) = 10 * I
  END DO

END
```

9

Παράδειγμα ανάθεσης τιμών #2

Δηλώστε πίνακα ακεραίων με 5 στοιχεία, και αναθέστε τους τις τιμές 10, 20, ..., 50 αντίστοιχα

2. Εντολή DATA:

```
PROGRAM MATRIX3B
  IMPLICIT NONE
  INTEGER A(5)

  DATA A / 10, 20, 30, 40, 50/

END
```

Προσοχή: στην εντολή DATA υποχρεωτικά αναθέτουμε όλες τις τιμές του πίνακα. Εάν θέλουμε να αναθέσουμε μέρος αυτών, χρησιμοποιούμε μια από τις άλλες δύο¹⁰

Παράδειγμα ανάθεσης τιμών #3 (1/3)

Δηλώστε πίνακα ακεραίων με 5 στοιχεία, και αναθέστε τους τις τιμές 10, 20, ..., 50 αντίστοιχα

3. Εντολή READ 1:

```
PROGRAM MATRIX3C
  IMPLICIT NONE
  INTEGER A(5)

  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ 5 ΤΙΜΕΣ'
  READ(*,*) A(1), A(2), A(3), A(4), A(5)

END
```

Η ανάθεση εδώ γίνεται από το πληκτρολόγιο¹¹

Παράδειγμα ανάθεσης τιμών #3 (2/3)

Δηλώστε πίνακα ακεραίων με 5 στοιχεία, και αναθέστε τους τις τιμές 10, 20, ..., 50 αντίστοιχα

3. Εντολή READ 2:

```
PROGRAM MATRIX3C
  IMPLICIT NONE
  INTEGER A(5)

  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ 5 ΤΙΜΕΣ'
  READ(*,*) A

END
```

Προσοχή: εδώ υποχρεωτικά αναθέτουμε όλα τα στοιχεία του A. Εάν θέλουμε να αναθέσουμε μέρος αυτών, χρησιμοποιούμε μια από τις άλλες δύο μεθόδους READ.¹²

Παράδειγμα ανάθεσης τιμών #3 (3/3)

Δηλώστε πίνακα ακεραίων με 5 στοιχεία, και αναθέστε τους τις τιμές 10, 20, ..., 50 αντίστοιχα

3. Εντολή READ 3:

```
PROGRAM MATRIX3C
  IMPLICIT NONE
  INTEGER A(5), I

  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ 5 ΤΙΜΕΣ'
  READ(*,*) (A(I), I = 1, 5)

END
```

Το παραπάνω ονομάζεται έμμεσος βρόγχος DO₁₃

Βρόγχοι DO και εντολή READ

Ποιά η διαφορά στα δύο παρακάτω παραδείγματα;

Έμμεσο DO
(το DO "μέσα"
στο READ)

```
PROGRAM MATRIX3C
  IMPLICIT NONE
  INTEGER A(5), I
  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ 5 ΤΙΜΕΣ'
  READ(*,*) (A(I), I = 1, 5)

END
```

Άμεσο DO
(το READ "μέσα"
στο DO)

```
PROGRAM MATRIX3C2
  IMPLICIT NONE
  INTEGER A(5), I
  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ 5 ΤΙΜΕΣ'
  DO I = 1, 5
    READ(*,*) A(I)
  END DO

END
```

14

Παράδειγμα εξαγωγής τιμών #1 (1/2)

Δηλώστε πίνακα ακεραίων με 5 στοιχεία, και εισάγετε 5 τιμές. Κατόπιν εξάγετε το τετράγωνο της κάθε τιμής.

```
PROGRAM MATRIX3C
  IMPLICIT NONE
  INTEGER A(5), I

  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ 5 ΤΙΜΕΣ'
  READ(*,*) (A(I), I = 1, 5)

  WRITE(*,*) 'ΤΑ ΤΕΤΡΑΓΩΝΑ ΤΩΝ ΑΡΙΘΜΩΝ ΕΙΝΑΙ:'
  WRITE(*,*) A(1)**2, A(2)**2, A(3)**2, A(4)**2, A(5)**2

END
```

15

Παράδειγμα εξαγωγής τιμών #1 (2/2)

Δηλώστε πίνακα ακεραίων με 5 στοιχεία, και εισάγετε 5 τιμές. Κατόπιν εξάγετε το τετράγωνο της κάθε τιμής.

```
PROGRAM MATRIX3C
  IMPLICIT NONE
  INTEGER A(5), I

  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ 5 ΤΙΜΕΣ'
  READ(*,*) (A(I), I = 1, 5)

  WRITE(*,*) 'ΤΑ ΤΕΤΡΑΓΩΝΑ ΤΩΝ ΑΡΙΘΜΩΝ ΕΙΝΑΙ:'
  WRITE(*,*) (A(I)**2, I = 1, 5)

END
```

Με έμμεσο βρόγχο DO

16

Βρόγχοι DO και εντολή WRITE

Ποιά η διαφορά στα δύο παρακάτω παραδείγματα;

Έμμεσο DO

```
WRITE(*,*) (A(I)**2, I = 1, 5)
```

Άμεσο DO

```
DO I = 1, 5
    WRITE(*,*) A(I)**2
END DO
```

17

Παράδειγμα #3

- Πρόγραμμα που να ζητάει την εισαγωγή δεδομένων και να υπολογίζει τον μέσο όρο

```
PROGRAM AVERAGE
    IMPLICIT NONE
    INTEGER NMAX, N, I
    PARAMETER(NMAX = 1000)
    DOUBLE PRECISION X(NMAX), A

    WRITE(*,*) 'ΠΟΣΕΣ ΤΙΜΕΣ ΘΑ ΕΙΣΑΓΕΤΕ'
    READ(*,*) N

    !.... ΕΛΕΓΧΟΥΜΕ ΝΑ ΜΗΝ ΞΕΠΕΡΑΣΤΟΥΝ ΤΑ ΟΡΙΑ ΤΟΥ ΠΙΝΑΚΑ
    IF (N > NMAX .OR. N <= 0) THEN
        WRITE(*,*) 'ΛΑΘΟΣ: ΜΕΧΡΙ', NMAX, 'ΤΙΜΕΣ'
        STOP
    END IF
```

18
ΣΥΝΕΧΙΖΕΤΑΙ...

Παράδειγμα #3 (συνέχεια...)

- Πρόγραμμα που να ζητάει την εισαγωγή δεδομένων και να υπολογίζει τον μέσο όρο

```
WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΙΣ ΤΙΜΕΣ'
READ(*,*) (X(I), I = 1, N)

A = 0
DO I = 1, N
    A = A + X(I)
END DO

A = A / N

WRITE(*,*) 'Ο ΜΕΣΟΣ ΟΡΟΣ ΤΩΝ ΣΤΟΙΧΕΙΩΝ ΕΙΝΑΙ:', A

END
```

19

Παράδειγμα #4

- Πρόγραμμα που να ζητάει την εισαγωγή δεδομένων και να υπολογίζει τον μικρότερο όρο

```
PROGRAM XMIN
    IMPLICIT NONE
    INTEGER NMAX, N, I
    PARAMETER(NMAX = 1000)
    DOUBLE PRECISION X(NMAX), XMIN

    WRITE(*,*) 'ΠΟΣΕΣ ΤΙΜΕΣ ΘΑ ΕΙΣΑΓΕΤΕ'
    READ(*,*) N

    !.... ΕΛΕΓΧΟΥΜΕ ΝΑ ΜΗΝ ΞΕΠΕΡΑΣΤΟΥΝ ΤΑ ΟΡΙΑ ΤΟΥ ΠΙΝΑΚΑ
    IF (N > NMAX .OR. N <= 0) THEN
        WRITE(*,*) 'ΛΑΘΟΣ: ΜΕΧΡΙ', NMAX, 'ΤΙΜΕΣ'
        STOP
    END IF
```

20
ΣΥΝΕΧΙΖΕΤΑΙ...

Παράδειγμα #4 (συνέχεια...)

- Πρόγραμμα που να ζητάει την εισαγωγή δεδομένων και να υπολογίζει τον μικρότερο όρο

```

WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΙΣ ΤΙΜΕΣ'
READ(*,*) (X(I), I = 1, N)

XMIN = X(1)
DO I = 2, N
    IF (X(I) < XMIN) XMIN = X(I)
END DO

WRITE(*,*) 'Ο ΜΙΚΡΟΤΕΡΟΣ ΟΡΟΣ ΕΙΝΑΙ:', XMIN

END

```

21

Παράδειγμα #5

- Πρόγραμμα που να ζητάει την εισαγωγή δεδομένων και να υπολογίζει τον μεγαλύτερο όρο και την θέση στον πίνακα στην οποία εμφανίζεται αυτός

```

PROGRAM XMAX
  IMPLICIT NONE
  INTEGER NMAX, N, I, IMAX
  PARAMETER (NMAX = 1000)
  DOUBLE PRECISION X(NMAX), XMAX

  WRITE(*,*) 'ΠΟΣΕΣ ΤΙΜΕΣ ΘΑ ΕΙΣΑΓΕΤΕ'
  READ(*,*) N

  !.... ΕΛΕΓΧΟΥΜΕ ΝΑ ΜΗΝ ΞΕΠΕΡΑΣΤΟΥΝ ΤΑ ΟΡΙΑ ΤΟΥ ΠΙΝΑΚΑ
  IF (N > NMAX .OR. N <= 0) THEN
    WRITE(*,*) 'ΛΑΘΟΣ: ΜΕΧΡΙ', NMAX, 'ΤΙΜΕΣ'
    STOP
  END IF

```

22
ΣΥΝΕΧΙΖΕΤΑΙ...

Παράδειγμα #5 (συνέχεια...)

- Πρόγραμμα που να ζητάει την εισαγωγή δεδομένων και να υπολογίζει τον μεγαλύτερο όρο και την θέση του

```

WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΙΣ ΤΙΜΕΣ'
READ(*,*) (X(I), I = 1, N)

XMAX = X(1)
IMAX = 1
DO I = 2, N
    IF (X(I) > XMAX) THEN
        XMAX = X(I)
        IMAX = I
    END IF
END DO

WRITE(*,*) 'Ο ΜΕΓΑΛΥΤΕΡΟΣ ΟΡΟΣ ΕΙΝΑΙ:', XMAX
WRITE(*,*) 'ΚΑΙ ΕΜΦΑΝΙΖΕΤΑΙ ΣΤΗΝ ΘΕΣΗ:', IMAX


```

23

END


Εύρεση βέλτιστης ευθείας

- Πολλές φορές χρειαζόμαστε να βρούμε την καλύτερη ευθεία που περιγράφει μια συλλογή πειραματικών σημείων
 - π.χ. πειραματικές μετρήσεις όπου x είναι η θερμοκρασία και y είναι η ηλεκτρική αντίσταση ενός μετάλλου



Εύρεση βέλτιστης ευθείας

- Στην βέλτιστη ευθεία ελαχιστοποιούνται οι κατακόρυφες αποστάσεις των σημείων από αυτήν
 - δηλ. ελαχιστοποιούνται τα λάθη στην πρόβλεψη τιμής



Η βέλτιστη ευθεία βρίσκεται με την Μέθοδο των Ελαχίστων Τετραγώνων

25

Παράδειγμα #6 Μέθοδος Ελαχίστων Τετραγώνων

- Η ευθεία που ελαχιστοποιεί τα σφάλματα (κατακόρυφες αποστάσεις) από N σημεία είναι η:

$$y = a + xb$$

$$a = \frac{s_{xx}s_y - s_x s_{xy}}{Ns_{xx} - s_x s_x}, \quad b = \frac{Ns_{xy} - s_x s_y}{Ns_{xx} - s_x s_x}$$

$$s_x = \sum_{i=1}^N X_i$$

$$s_y = \sum_{i=1}^N Y_i$$

$$s_{xx} = \sum_{i=1}^N X_i^2$$

$$s_{xy} = \sum_{i=1}^N X_i Y_i$$

Παράδειγμα #6 Μέθοδος Ελαχίστων Τετραγώνων (1/2)

```
PROGRAM LEAST_SQUARES
  IMPLICIT NONE
  INTEGER NMAX, N, I
  PARAMETER (NMAX = 1000)
  DOUBLE PRECISION X(NMAX), Y(NMAX)
  DOUBLE PRECISION SX, SY, SXX, SXY, A, B

  WRITE(*,*) 'ΠΟΣΑ ΣΗΜΕΙΑ ΘΑ ΕΙΣΑΓΕΤΕ;'
  READ(*,*) N
  IF (N > NMAX .OR. N <= 1) THEN
    WRITE(*,*) 'ΛΑΘΟΣ: ΜΕΧΡΙ', NMAX, 'ΤΙΜΕΣ'
    STOP
  END IF

  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΑ ΣΗΜΕΙΑ'
  DO I = 1, N
    READ(*,*) X(I), Y(I)
  END DO
```

27
ΣΥΝΕΧΙΖΕΤΑΙ...

Παράδειγμα #6 Μέθοδος Ελαχίστων Τετραγώνων (2/2)

```
SX = 0
SY = 0
SXX = 0
SXY = 0

DO I = 1, N
  SX = SX + X(I)
  SY = SY + Y(I)
  SXX = SXX + X(I)**2
  SXY = SXY + X(I) * Y(I)
END DO

A = (SXX*SY - SX*SXY) / (N*SXX - SX*SX)
B = (N*SXY - SX*SY) / (N*SXX - SX*SX)

WRITE(*,*) 'Η ΚΛΙΣΗ ΤΗΣ ΕΥΘΕΙΑΣ ΕΙΝΑΙ:', A
WRITE(*,*) 'ΚΑΙ Ο ΣΤΑΘΕΡΟΣ ΤΗΣ ΟΡΟΣ:', B

END
```

28

Παράδειγμα #7 Εσωτερικό γινόμενο δύο διανυσμάτων

- Στον τριδιάστατο χώρο, για δύο διανύσματα

$$R_1 = (x_1, y_1, z_1) \quad R_2 = (x_2, y_2, z_2)$$

- Το εσωτερικό τους γινόμενο ορίζεται ως

$$R_1 \cdot R_2 = x_1 x_2 + y_1 y_2 + z_1 z_2$$

- Στην γενική περίπτωση διανυσμάτων N διαστάσεων

$$X = (x_1, x_2, x_3, \dots, x_N) \quad Y = (y_1, y_2, y_3, \dots, y_N)$$

- Το εσωτερικό τους γινόμενο ορίζεται ως

$$X \cdot Y = x_1 y_1 + x_2 y_2 + x_3 y_3 + \dots + x_N y_N \quad 29$$

Παράδειγμα #7 Εσωτερικό γινόμενο δύο διανυσμάτων

```
WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΙΣ ΤΙΜΕΣ ΤΟΥ 1ΟΥ ΔΙΑΝΥΣΜΑΤΟΣ'
READ(*,*) (X(I), I = 1, N)
WRITE(*,*) 'ΤΩΡΑ ΤΙΣ ΤΙΜΕΣ ΤΟΥ 2ΟΥ ΔΙΑΝΥΣΜΑΤΟΣ'
READ(*,*) (Y(I), I = 1, N)
```

```
PRODUCT = 0
DO I = 1, N
    PRODUCT = PRODUCT + X(I) * Y(I)
END DO
```

```
WRITE(*,*) 'ΤΟ ΕΣΩΤΕΡΙΚΟ ΓΙΝΟΜΕΝΟ ΕΙΝΑΙ', PRODUCT
```

END

Παράδειγμα #7 Εσωτερικό γινόμενο δύο διανυσμάτων

Κατασκευάστε πρόγραμμα που δέχεται δύο διανύσματα N διαστάσεων και θα υπολογίζει το εσωτερικό τους γινόμενο

```
PROGRAM DOT_PRODUCT
  IMPLICIT NONE
  INTEGER NMAX, N, I
  PARAMETER (NMAX = 1000)
  DOUBLE PRECISION X(NMAX), Y(NMAX), PRODUCT


  WRITE(*,*) 'ΠΟΣΑ ΔΙΑΣΤΑΣΕΙΣ ΕΙΝΑΙ ΤΑ ΔΙΑΝΥΣΜΑΤΑ;'
  READ(*,*) N

  IF (N > NMAX .OR. N <= 0) THEN
    WRITE(*,*) 'ΛΑΘΟΣ: ΜΕΧΡΙ', NMAX, 'ΤΙΜΕΣ'
    STOP
  END IF
```

ΣΥΝΕΧΙΖΕΤΑΙ³⁰ .

Παράδειγμα #8 Αριθμός σημείων μέσα σε κύκλο

- Έστω N σημεία στο επίπεδο $\{(x_i, y_i), \text{ για } i=1, 2, \dots, N\}$
Πόσα είναι μέσα σε κύκλο ακτίνας r και κέντρου (x_0, y_0) ;



απόσταση σημείου από
το κέντρο του κύκλου:

$$d_i = \sqrt{(x_i - x_0)^2 + (y_i - y_0)^2}$$

το σημείο είναι μέσα
στον κύκλο εαν

$$d_i \leq r$$

Παράδειγμα #8 Αριθμός σημείων μέσα σε κύκλο (1/2)

Κατασκευάστε πρόγραμμα που δέχεται N σημεία και υπολογίζει πόσα από αυτά ανήκουν σε δοθέν κύκλο

```
PROGRAM POINTS_IN_CIRCLE
  IMPLICIT NONE
  INTEGER NMAX, N, I, NUMBER
  PARAMETER(NMAX = 1000)
  DOUBLE PRECISION X(NMAX), Y(NMAX), X0, Y0, R, D

  WRITE(*,*) 'ΠΟΣΑ ΣΗΜΕΙΑ ΘΑ ΕΙΣΑΓΕΤΕ;'
  READ(*,*) N

  IF (N > NMAX .OR. N <= 0) THEN
    WRITE(*,*) 'ΛΑΘΟΣ: ΜΕΧΡΙ', NMAX, 'ΤΙΜΕΣ'
    STOP
  END IF
```

33
ΣΥΝΕΧΙΖΕΤΑΙ...

Παράδειγμα #8 Αριθμός σημείων μέσα σε κύκλο (2/2)

```
WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΑ ΣΗΜΕΙΑ'
DO I = 1, N
  READ(*,*) X(I), Y(I)
END DO

WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΟΝ ΚΥΚΛΟ'
READ(*,*) X0, Y0, R

NUMBER = 0
DO I = 1, N
  D = SQRT( (X(I)-X0)**2 + (Y(I)-Y0)**2 )
  IF (D <= R) NUMBER = NUMBER + 1
END DO

WRITE(*,*) 'ΜΕΣΑ ΣΤΟΝ ΚΥΚΛΟ ΕΙΝΑΙ', NUMBER, 'ΣΗΜΕΙΑ'
```

END

34

Παράδειγμα #9 Αναδρομική ακολουθία

Κατασκευάστε πρόγραμμα που δέχεται έναν αριθμό και αναθέτει σε πίνακα τις 100 πρώτες τιμές της παρακάτω ακολουθίας. Κατόπιν να τυπώνει τους 50 πρώτους όρους

$$X_{n+1} = X_n (1 - X_n)$$

ο αριθμός που δέχεται το πρόγραμμα είναι ο πρώτος όρος της ακολουθίας

35

Παράδειγμα #9 Αναδρομική ακολουθία

```
PROGRAM ANADROMIKI
  IMPLICIT NONE
  INTEGER NMAX, N, I
  PARAMETER(NMAX = 100)
  DOUBLE PRECISION X(NMAX)

  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΕΝΑΝ ΑΡΙΘΜΟ'
  READ(*,*) X(1)

  DO I = 2, 100
    X(I) = X(I-1) * (1 - X(I-1))
  END DO

  DO I = 1, 50
    WRITE(*,*) 'Ο ΟΡΟΣ', I, 'ΕΙΝΑΙ', X(I)
  END DO
```

END

36

Παράδειγμα #10 Στατιστική ανάλυση

- Σε ένα σύνολο μετρήσεων $x_i, i=1, N$, υπολογίζουμε

μέσος όρος

$$\bar{X} = \frac{\sum_{i=1}^N x_i}{N}$$

στατιστική απόκλιση (σφάλμα)

$$\sigma_x = \left[\frac{\sum_{i=1}^N (x_i - \bar{X})^2}{N-1} \right]^{1/2}$$

σφάλμα μέσης τιμής

$$\sigma_{\bar{X}} = \frac{\sigma_x}{\sqrt{N}}$$

σχετικό σφάλμα

$$r = \frac{\sigma_{\bar{X}}}{\bar{X}} \cdot 100$$

αποτέλεσμα: $x = \bar{x} \pm \sigma_{\bar{x}}$

37

Παράδειγμα #10 Στατιστική ανάλυση (1/2)

Κατασκευάστε πρόγραμμα που δέχεται N αριθμούς και θα κάνει στατιστική ανάλυση

```
PROGRAM STATISTICS
  IMPLICIT NONE
  INTEGER NMAX, N, I
  PARAMETER (NMAX = 1000)
  DOUBLE PRECISION X(NMAX), XMEAN, STDY, STDY1, ERROR

  WRITE(*,*) 'ΠΟΣΕΣ ΤΙΜΕΣ ΘΑ ΕΙΣΑΓΕΤΕ'
  READ(*,*) N
  IF (N > NMAX .OR. N <= 0) THEN
    WRITE(*,*) 'ΛΑΘΟΣ: ΜΕΧΡΙ', NMAX, 'ΤΙΜΕΣ'
    STOP
  END IF
  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΙΣ ΤΙΜΕΣ'
  READ(*,*) (X(I), I = 1, N)
  ΣΥΝΕΧΙΖΕΤΑΙ38..
```

Παράδειγμα #10 Στατιστική ανάλυση (2/2)

```
XMEAN = 0
DO I = 1, N
  XMEAN = XMEAN + X(I)
END DO
XMEAN = XMEAN / N
```

$$\bar{X} = \frac{\sum_{i=1}^N x_i}{N}$$

```
STDY = 0
DO I = 1, N
  STDY = STDY + (X(I) - XMEAN)**2
END DO
STDY = SQRT(STDY / (N-1))
```

$$\sigma_x = \left[\frac{\sum_{i=1}^N (x_i - \bar{X})^2}{N-1} \right]^{1/2}$$

```
STDY1 = STDY / SQRT(N)
R = STDY1 / XMEAN * 100
```

$$\sigma_{\bar{X}} = \frac{\sigma_x}{\sqrt{N}} \quad r = \frac{\sigma_{\bar{X}}}{\bar{X}} \cdot 100$$

```
WRITE(*,*) 'ΤΟ ΑΠΟΤΕΛΕΣΜΑ ΕΙΝΑΙ', XMEAN, '+-', STDY1
WRITE(*,*) 'ΚΑΙ ΕΜΦΑΝΙΖΕΙ ΣΧΕΤΙΚΟ ΣΦΑΛΜΑ', R, '%'
39
```

END

Παράδειγμα #11 Ταξινόμηση στοιχείων πίνακα

- Είναι πολύ συχνό σε εφαρμογές να θέλουμε να ταξινομήσουμε τα στοιχεία ενός πίνακα
 - π.χ. ταξινόμηση σε αύξουσα σειρά

Έστω ο πίνακας X με στοιχεία

$X = \{13, 25, 6, 94, 26, 73, 14, 59, 93, 65\}$

συγκρίνουμε το πρώτο με το δεύτερο:
εάν μεγαλύτερο, τους αλλάζουμε θέση

συγκρίνουμε το δεύτερο με το τρίτο:
εάν μεγαλύτερο, τους αλλάζουμε θέση

. KOK

στο τέλος του πρώτου κύκλου, το μεγαλύτερο στοιχείο είναι στη τελευταία θέση του πίνακα

40

Παράδειγμα #11 Ταξινόμηση στοιχείων πίνακα

Έστω ο πίνακας X με 10 στοιχεία

$X = \{13, 25, 6, 94, 26, 73, 14, 59, 93, 65\}$

εκτελούμε τον πρώτο κύκλο με N βήματα
 $X = \{13, 6, 25, 26, 73, 14, 59, 93, 65, 94\}$

εκτελούμε τον δεύτερο κύκλο με N-1 βήματα
 $X = \{6, 13, 25, 26, 14, 59, 73, 65, 93, 94\}$

εκτελούμε τον τρίτο κύκλο με N-2 βήματα
 $X = \{6, 13, 25, 14, 26, 59, 65, 73, 93, 94\}$

.
KOK

στο τέλος N-1
κύκλων, όλα τα
στοιχεία είναι
ταξινομημένα

41

Παράδειγμα #11 Ταξινόμηση στοιχείων πίνακα (1/2)

Κατασκευάστε πρόγραμμα που δέχεται έναν πίνακα ρητών και τους ταξινομεί κατά αύξουσα σειρά

```
PROGRAM BUBBLE_SORT
  IMPLICIT NONE
  INTEGER NMAX, N, I, J
  PARAMETER (NMAX = 1000)
  DOUBLE PRECISION X(NMAX), TEMP

  WRITE(*,*) 'ΠΟΣΕΣ ΤΙΜΕΣ ΘΑ ΕΙΣΑΓΕΤΕ'
  READ(*,*) N
  IF (N > NMAX .OR. N <= 0) THEN
    WRITE(*,*) 'ΛΑΘΟΣ: ΜΕΧΡΙ', NMAX, 'ΤΙΜΕΣ'
    STOP
  END IF
  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΙΣ ΤΙΜΕΣ'
  READ(*,*) (X(I), I = 1, N)
  ΣΥΝΕΧΙΖΕΤΑΙ...42
```

Παράδειγμα #11 Ταξινόμηση στοιχείων πίνακα (2/2)

```
!.....ΓΙΑ ΚΑΘΕ ΘΕΣΗ J
  DO J = N, 2, -1
!.....ΒΡΙΣΚΟΥΜΕ ΚΑΙ ΤΟΠΟΘΕΤΟΥΜΕ
!.....ΤΟ ΑΝΤΙΣΤΟΙΧΟ ΜΕΓΙΣΤΟ
    DO I = 1, J-1
      IF (X(I) > X(I+1)) THEN
        TEMP = X(I)
        X(I) = X(I+1)
        X(I+1) = TEMP
      END IF
    END DO
  END DO

  WRITE(*,*) 'Ο ΤΑΞΙΝΟΜΗΜΕΝΟΣ ΠΙΝΑΚΑΣ ΕΙΝΑΙ:'
  WRITE(*,*) (X(I), I = 1, N)
```

END

43

Παράδειγμα #12 Διαδική αναζήτηση

Δεδομένου ταξινομημένου πίνακα, να φτιάξετε πρόγραμμα που δέχεται έναν αριθμό και βρίσκει την κοντινότερη θέση στον πίνακα

παράδειγμα, έστω πίνακας
 $X = \{2, 5, 8, 12, 16, 32, 43, 47, 76, 78, 84, 91, 92\}$
και μας ζητείται να βρούμε την κοντινότερη θέση του 90

εαν τα ψάξουμε ένα-ένα, θα πάρει πολύ χρόνο

Γρήγορος τρόπος:

Κοιτάμε στην μέση του πίνακα και αποφασίζουμε εαν το νούμερο είναι απο τα αριστερά ή απο τα δεξιά
Θέτουμε νέα όρια στον πίνακα
Κοιτάμε στην νέα μέση του πίνακα και αποφασίζουμε...

44

Παράδειγμα #12

Δυαδική αναζήτηση (1/2)

```
PROGRAM BINARY_SEARCH
  IMPLICIT NONE
  INTEGER NMAX, N, I1, I2, IC
  PARAMETER(NMAX = 1000)
  DOUBLE PRECISION X(NMAX), X0

  WRITE(*,*) 'ΔΩΣΤΕ ΕΝΑ ΝΟΥΜΕΡΟ'
  READ(*,*) X0
  I1 = 1
  I2 = N
  DO WHILE (I2-I1 > 1)
    IC = (I1 + I2) / 2
    IF (X0 > X(IC)) THEN
      I1 = IC
    ELSE
      I2 = IC
    END IF
  END DO
  END DO
```

ΣΥΝΕΧΙΖΕΤΑΙ...

45

Παράδειγμα #12

Δυαδική αναζήτηση (2/2)


```
IF ( X0-X(I1) < X(I2)-X0 ) THEN
  WRITE(*,*) 'Η ΚΟΝΤΙΝΟΤΕΡΗ ΘΕΣΗ ΕΙΝΑΙ Η', I1
ELSE
  WRITE(*,*) 'Η ΚΟΝΤΙΝΟΤΕΡΗ ΘΕΣΗ ΕΙΝΑΙ Η', I2
END IF
END
```

46

Παράδειγμα #13

Ξεδιάλεγμα σημείων μέσα σε κύκλο

- Έστω N σημεία στο επίπεδο $\{ (x_i, y_i), \text{ για } i=1, 2, \dots, N \}$
Βρείτε και κρατήστε μόνο όσα είναι μέσα σε κύκλο ακτίνας r και κέντρου (x_0, y_0)



απόσταση σημείου από
το κέντρο του κύκλου:

$$d_i = \sqrt{(x_i - x_0)^2 + (y_i - y_0)^2}$$

το σημείο είναι μέσα
στον κύκλο εαν

$$d_i \leq r$$

47

Παράδειγμα #13

Ξεδιάλεγμα σημείων μέσα σε κύκλο(1/2)

Κατασκευάστε πρόγραμμα που δέχεται N σημεία και κρατάει μόνο όσα ανήκουν σε δοθέν κύκλο

```
PROGRAM POINTS_IN_CIRCLE2
  IMPLICIT NONE
  INTEGER NMAX, N, I, INEW
  PARAMETER(NMAX = 1000)
  DOUBLE PRECISION X(NMAX), Y(NMAX), X0, Y0, R, D

  WRITE(*,*) 'ΠΟΣΑ ΣΗΜΕΙΑ ΘΑ ΕΙΣΑΓΕΤΕ;'
  READ(*,*) N

  IF (N > NMAX .OR. N <= 0) THEN
    WRITE(*,*) 'ΛΑΘΟΣ: ΜΕΧΡΙ', NMAX, 'ΤΙΜΕΣ'
    STOP
  END IF
```

ΣΥΝΕΧΙΖΕΤΑΙ...

48

Παράδειγμα #13

Ξεδιάλεγμα σημείων μέσα σε κύκλο(2/2)

```

WRITE(*,*) `ΕΙΣΑΓΕΤΕ ΤΑ ΣΗΜΕΙΑ`
READ(*,*) (X(I), I = 1, N), (Y(I), I = 1, N)
WRITE(*,*) `ΕΙΣΑΓΕΤΕ ΤΟΝ ΚΥΚΛΟ`
READ(*,*) X0, Y0, R

INEW = 0
DO I = 1, N
    D = SQRT( (X(I)-X0)**2 + (Y(I)-Y0)**2 )
    IF (D <= R) THEN
        INEW = INEW + 1
        X(INEW) = X(I)
        Y(INEW) = Y(I)
    END IF
END DO
N = INEW

```

END

49

Παράδειγμα #14

Ταξινόμηση στοιχείων πίνακα (απέυθειας επιλογή)

Έστω ο πίνακας X με 10 στοιχεία

$X = \{13, 25, 6, 94, 26, 73, 14, 59, 93, 65\}$

Βρίσκουμε το μικρότερο και το βάζουμε στην αρχή
 $X = \{6, 25, 13, 94, 26, 73, 14, 59, 93, 65\}$

Βρίσκουμε το επόμενο μικρότερο και το βάζουμε στην δεύτερη θέση
 $X = \{6, 13, 25, 94, 26, 73, 14, 59, 93, 65\}$

Βρίσκουμε το επόμενο μικρότερο και το βάζουμε στην δεύτερη θέση
 $X = \{6, 13, 14, 94, 26, 73, 25, 59, 93, 65\}$

. ΚΟΚ

στο τέλος N-1 κύκλων, όλα τα στοιχεία είναι ταξινομημένα

50

Παράδειγμα #14

Ταξινόμηση στοιχείων πίνακα (απέυθειας επιλογή) (1/2)

Κατασκευάστε πρόγραμμα που δέχεται έναν πίνακα ρητών και τους ταξινομεί κατά αύξουσα σειρά

```

PROGRAM MIN_SORT
IMPLICIT NONE
INTEGER NMAX, N, I, J, IMIN
PARAMETER(NMAX = 1000)
DOUBLE PRECISION X(NMAX), TEMP

WRITE(*,*) `ΠΟΣΕΣ ΤΙΜΕΣ ΘΑ ΕΙΣΑΓΕΤΕ`
READ(*,*) N
IF (N > NMAX .OR. N <= 0) THEN
    WRITE(*,*) `ΛΑΘΟΣ: ΜΕΧΡΙ`, NMAX, `ΤΙΜΕΣ`
    STOP
END IF
WRITE(*,*) `ΕΙΣΑΓΕΤΕ ΤΙΣ ΤΙΜΕΣ`
READ(*,*) (X(I), I = 1, N)

```

ΣΥΝΕΧΙΖΕΤΑΙ...⁵¹

Παράδειγμα #14

Ταξινόμηση στοιχείων πίνακα (απέυθειας επιλογή) (2/2)

```


!.....ΓΙΑ ΚΑΘΕ ΘΕΣΗ J
DO J = 1, N-1
!.....ΒΡΙΣΚΟΥΜΕ ΤΟ ΑΝΤΙΣΤΟΙΧΟ ΕΛΑΧΙΣΤΟ
    XMIN = X(J)
    IMIN = J
    DO I = J+1, N
        IF (X(I) < XMIN) THEN
            XMIN = X(I)
            IMIN = I
        END IF
    END DO
    TEMP = X(J)
    X(J) = X(IMIN)
    X(IMIN) = TEMP
END DO
WRITE(*,*) `Ο ΤΑΞΙΝΟΜΗΜΕΝΟΣ ΠΙΝΑΚΑΣ ΕΙΝΑΙ:`
WRITE(*,*) (X(I), I = 1, N)
END

```

52


Παράδειγμα #15 Μετατόπιση σημείων και στροφή

- Έστω N σημεία στο επίπεδο $\{ (x_i, y_i), \text{ για } i=1, 2, \dots, N \}$
Μετακινήστε τα στη μέση θέση τους, και μετά στρέψτε τα κατά γωνία θ



53

Στροφή διανύσματος κατά γωνία θ



$$|R| = \sqrt{x_1^2 + y_1^2}$$

$$|R| = |R'|$$

$$\theta_1 = \arctan(y_1 / x_1)$$

$$\theta_2 = \theta_1 + \theta$$

$$x_2 = |R| \cos \theta_2 \quad y_2 = |R| \sin \theta_2$$

54

Παράδειγμα #15 Μετατόπιση σημείων και στροφή(1/4)

```
PROGRAM TURN_POINTS
  IMPLICIT NONE
  INTEGER NMAX, N, I
  PARAMETER (NMAX = 1000)
  DOUBLE PRECISION X(NMAX), Y(NMAX), THETA(NMAX), &
    THETA0, R, XMEAN, YMEAN, PI

  WRITE(*,*) 'ΠΟΣΑ ΣΗΜΕΙΑ ΘΑ ΕΙΣΑΓΕΤΕ;'
  READ(*,*) N
  IF (N > NMAX .OR. N <= 0) THEN
    WRITE(*,*) 'ΛΑΘΟΣ: ΜΕΧΡΙ', NMAX, 'ΤΙΜΕΣ'
    STOP
  END IF
  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΑ ΣΗΜΕΙΑ'
  READ(*,*) (X(I), I = 1, N), (Y(I), I = 1, N)
  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΗΝ ΓΩΝΙΑ ΣΤΡΟΦΗΣ ΣΕ MOIRES'
  READ(*,*) THETA0
  ΣΥΝΕΧΙΖΕΤΑΙ...
```

55

Παράδειγμα #15 Μετατόπιση σημείων και στροφή(2/4)

```
!.....ΒΡΙΣΚΟΥΜΕ ΤΗΝ ΜΕΣΗ ΘΕΣΗ
XMEAN = 0
YMEAN = 0
DO I = 1, N
  XMEAN = XMEAN + X(I)
  YMEAN = YMEAN + Y(I)
END DO

!.....ΜΕΤΑΚΙΝΟΥΜΕ ΤΑ ΣΗΜΕΙΑ ΚΑΤΑ ΤΗ ΜΕΣΗ ΘΕΣΗ ΤΟΥΣ
DO I = 1, N
  X(I) = X(I) - XMEAN / N
  Y(I) = Y(I) - YMEAN / N
END DO

PI = ACOS(-1.0)
ΣΥΝΕΧΙΖΕΤΑΙ...
```

56

Παράδειγμα #15

Μετατόπιση σημείων και στροφή(3/4)

```

!.....ΒΡΙΣΚΟΥΜΕ ΤΗΝ ΓΩΝΙΑ ΤΟΥ ΚΑΘΕ ΣΗΜΕΙΟΥ
DO I = 1, N
  IF (X(I) == 0 .AND. Y(I) == 0) THEN
    THETA(I) = 0
  ELSE IF (X(I) == 0 .AND. Y(I) > 0) THEN
    THETA(I) = PI / 2
  ELSE IF (X(I) == 0 .AND. Y(I) < 0) THEN
    THETA(I) = 3 * PI / 2
  ELSE IF (X(I) < 0) THEN
    THETA(I) = ATAN( Y(I) / X(I) ) + PI
  ELSE IF (Y(I) < 0) THEN
    THETA(I) = ATAN( Y(I) / X(I) ) + 2 * PI
  ELSE
    THETA(I) = ATAN( Y(I) / X(I) )
  END IF
END DO

```

57
ΣΥΝΕΧΙΖΕΤΑΙ...

Παράδειγμα #15

Μετατόπιση σημείων και στροφή(4/4)

```

!.....ΣΤΡΕΦΟΥΜΕ ΤΟ ΔΙΑΝΥΣΜΑ ΤΟΥ ΚΑΘΕ ΣΗΜΕΙΟΥ
DO I = 1, N
  THETA(I) = THETA(I) + THETA0 * PI / 180
  R = SQRT ( X(I)**2 + Y(I)**2 )
  X(I) = R * COS (THETA(I) )
  Y(I) = R * SIN (THETA(I) )
END DO
END


```

58

Παράδειγμα #16

Κέντρο μάζας ατόμων και μέση απόσταση

Κατασκευάστε πρόγραμμα που δέχεται N άτομα (2 διαστάσεις) και τις μάζες τους και υπολογίζει το κέντρο μάζας και την μέση απόσταση μεταξύ τους



Κέντρο μάζας

$$\mu_x = \frac{\sum_{i=1}^N x_i m_i}{\sum_{i=1}^N m_i}$$

Μέση απόσταση

$$\bar{d} = \frac{\sum_{j=1}^N \sum_{i=j+1}^N d_{ij}}{N(N-1)/2}$$

59

Παράδειγμα #16

Κέντρο μάζας ατόμων και μέση απόσταση (1/3)

```

PROGRAM MASS_CENTER
  IMPLICIT NONE
  INTEGER NMAX, N, I
  PARAMETER (NMAX = 1000)
  DOUBLE PRECISION X(NMAX), Y(NMAX), &
    M(NMAX), MX, MY, MMEAN, D, DMEAN

  WRITE(*,*) 'ΠΟΣΑ ΑΤΟΜΑ ΘΑ ΕΙΣΑΓΕΤΕ;'
  READ(*,*) N
  IF (N > NMAX .OR. N <= 0) THEN
    WRITE(*,*) 'ΛΑΘΟΣ: ΜΕΧΡΙ', NMAX, 'ΤΙΜΕΣ'
    STOP
  END IF
  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΑ ΣΗΜΕΙΑ ΚΑΙ ΜΑΖΕΣ'
  READ(*,*) (X(I), I = 1, N), (Y(I), I = 1, N), &
    (M(I), I = 1, N)

```

60
ΣΥΝΕΧΙΖΕΤΑΙ...

Παράδειγμα #16 Κέντρο μάζας ατόμων και μέση απόσταση (2/3)

```
!.....ΒΡΙΣΚΟΥΜΕ ΤΟ ΚΕΝΤΡΟ ΜΑΖΑΣ
MX = 0
MY = 0
MMEAN = 0
DO I = 1, N
    MX = MX + X(I) * M(I)
    MY = MY + Y(I) * M(I)
    MMEAN = MMEAN + M(I)
END DO

MX = MX / MMEAN
MY = MY / MMEAN


WRITE(*,*) 'ΤΟ ΚΕΝΤΡΟ ΜΑΖΑΣ ΕΙΝΑΙ', MX, MY

ΣΥΝΕΧΙΖΕΤΑΙ...
```


$$\mu_x = \frac{\sum_{i=1}^N x_i m_i}{\sum_{i=1}^N m_i}$$

Παράδειγμα #17 Ιστόγραμμα απόστασης ατόμων

Κατασκευάστε πρόγραμμα που δέχεται N άτομα (2 διαστάσεις) και υπολογίζει τις σχετικές τους αποστάσεις και δημιουργεί ιστόγραμμα με την κατανομή τους



Για παράδειγμα, έστω ότι θέλουμε να κατασκευάσουμε ένα ιστόγραμμα με 100 σημεία για αποστάσεις μέχρι DMAX. Διαιρούμε σε 100 διαστήματα πάχους H = DMAX / 100. Βρίσκουμε πόσες αποστάσεις είναι μέσα στο πρώτο διάστημα, πόσες στο δεύτερο, κ.ο.κ.



Παράδειγμα #16 Κέντρο μάζας ατόμων και μέση απόσταση (3/3)

```
!.....ΒΡΙΣΚΟΥΜΕ ΤΗ ΜΕΣΗ ΑΠΟΣΤΑΣΗ
DMEAN = 0
DO J = 1, N
    DO I = J+1, N
        D = SQRT((X(J)-X(I))**2+(Y(J)-Y(I))**2)
        DMEAN = DMEAN + D
    END DO
END DO

DMEAN = DMEAN / (N * (N-1) / 2)

WRITE(*,*) 'Η ΜΕΣΗ ΑΠΟΣΤΑΣΗ ΕΙΝΑΙ', DMEAN

END
```

$$\bar{d} = \frac{\sum_{j=1}^N \sum_{i=j+1}^N d_{ij}}{N(N-1)/2}$$

Παράδειγμα #17 Ιστόγραμμα απόστασης ατόμων(1/2)

```
PROGRAM HISTOGRAM
IMPLICIT NONE
INTEGER NMAX, N, I
PARAMETER(NMAX = 1000)
DOUBLE PRECISION X(NMAX), Y(NMAX), H, DMAX, D
INTEGER HIST(100)

WRITE(*,*) 'ΠΟΣΑ ΑΤΟΜΑ ΘΑ ΕΙΣΑΓΕΤΕ;'
READ(*,*) N
IF (N > NMAX .OR. N <= 0) THEN
    WRITE(*,*) 'ΛΑΘΟΣ: ΜΕΧΡΙ', NMAX, 'ΤΙΜΕΣ'
    STOP
END IF
WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΑ ΣΗΜΕΙΑ'
READ(*,*) (X(I), I = 1, N), (Y(I), I = 1, N)
WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΟ ΟΡΙΟ ΤΟΥ ΙΣΤΟΓΡΑΜΜΑΤΟΣ'
READ(*,*) DMAX
ΣΥΝΕΧΙΖΕΤΑΙ...
```


Παράδειγμα #17

Ιστόγραμμα απόστασης ατόμων(2/2)

```
!.....ΤΟ ΔΙΑΣΤΗΜΑ ΤΟΥ ΙΣΤΟΓΡΑΜΜΑΤΟΣ
H = DMAX / 100
DO K = 1, 100
    HIST(K) = 0
END DO

DO J = 1, N
    DO I = J+1, N
        D = SQRT((X(J)-X(I))**2+(Y(J)-Y(I))**2)
        K = INT(D/H) + 1
        IF (K < 100) HIST(K) = HIST(K) + 1
    END DO
END DO
WRITE(*,*) 'ΤΟ ΙΣΤΟΓΡΑΜΜΑ ΚΑΤΑΝΟΜΗΣ ΑΠΟΣΤΑΣΕΩΝ:'
DO K = 1, 100
    WRITE(*,*) 'ΑΠΟ', (K-1)*H, 'ΜΕΧΡΙ', K*H, HIST(K)
END DO
END
```